
帰納的な定義と自己反映
1 帰納的な定義
Coqの帰納的データ型
前回は自然数の定義を見た．

Inductive nat : Set := O : nat | S : nat -> nat.

実は，Coqの全てのデータは帰納的データ型として定義される．1

Inductive prod (A B : Set) : Set := pair : A -> B -> prod A B.

Inductive sum (A B : Set) : Set := inl : A -> sum A B | inr : B -> sum A B.

帰納的データ型の値を作るのは構成を適用するだけでいい．しかし，分解するのに OCamlと
同様にパターンマッチングを使わなければならない．その型付け規則が複雑になる．以下のよう
なデータ型を考える．

Inductive t(a1 . . . an : Set) : Set :=

| c1 : τ11 → . . . → τ1k1 → t a1 . . . an

. . .

| cm : τm1 → . . . → τmkm → t a1 . . . an.

マッチング
Γ ⊢ M : t b1 . . . bn

Γ, xi1 : τi1[b1/a1, . . . , bn/an], . . . , xiki : τiki [. . .] ⊢ Mi : τ [ci xi1...xiki/x] (1 ≤ i ≤ m)

Γ ⊢ match M as x return τ with c1 x11 . . . x1k1 ⇒ M1 | . . . | cm xm1 . . . xmkm ⇒ Mm end : τ [M/x]

asと returnによって，返り値の型の中に入力を含めることができ，場合によって型が違うよう
な関数が作れる．それを手でやるのは難しいが，作戦 caseはこのパターンマッチングを構築し
てくれる．
帰納的データ型を定義すると，帰納法のための補題が自動的に定義されるが，定義は match

を使う．定義が再帰的でないとき，パターンマッチングだけで済む．再帰的なデータ型について
Fixpointが使われる．
Definition prod_ind (A B:Set) (P:prod A B -> Prop) :=

fun (f : forall a b, P (pair a b)) =>

fun p => match p as x return P x with pair a b => f a b end.

Check prod_ind.

: forall (A B : Set) (P : A * B -> Prop),

(forall (a : A) (b : B), P (a, b)) -> forall p : A * B, P p

Definition sum_ind (A B:Set) (P:sum A B -> Prop) :=

1実際の定義を見ると，Setではなく Typeになっている．Typeは Setより一般的なので，Setとして使うことが
できる．さらに，prod A B は　 A*Bとして表示され，pair a bは (a,b)として表示される．Coqの Notationとい
う機能によって，機能的データ型の表示方法を変えることができる．

fun (fl : forall a, P (inl _ a)) (fr : forall b, P (inr _ b)) =>

fun p => match p as x return P x

with inl a => fl a | inr b => fr b end.

Check sum_ind.

: forall (A B : Set) (P : A + B -> Prop),

(forall a : A, P (inl B a)) -> (forall b : B, P (inr A b)) ->

forall p : A + B, P p

Fixpoint nat_ind (P:nat -> Prop) (f0:P O) (fn:forall n, P n -> P (S n))

(n : nat) {struct n} :=

match n as x return P x

with O => f0 | S m => fn m (nat_ind P f0 fn m) end.

Check nat_ind.

: forall P : nat -> Prop, P 0 -> (forall n : nat, P n -> P (S n)) ->

forall n : nat, P n

caseと elimはよく似ているが，前者が単なる場合分けを行うのに対して，後者が生成された
補題を利用しているので，効果が違ったりする．

Lemma plusn0 n : n + 0 = n.

Proof.

case: n.

+ done.

forall n : nat, S n + 0 = S n

Restart.

move: n.

apply: nat_ind. (* elim の意味 *)

+ done.

forall n : nat, n + 0 = n -> S n + 0 = S n

+ move=> n /= -> //.

Qed.

帰納的述語
Coqでは帰納的な定義は SetだけでなくPropでもできる．この場合，パラメータは場合によっ

て変わることが多い．

Inductive t : τ1 → . . . → τn → Prop :=

| c1 : ∀(x1 : τ11) . . . (xk1 : τ1k1), t θ11 . . . θ1n

. . .

| cm : ∀(x1 : τm1) . . . (xk1 : τmkm), t θm1 . . . θmn

マッチング
Γ ⊢ M : t b1 . . . bn Γ, xi1 : τi1, . . . , xiki : τiki ⊢ Mi : τ [θi1 . . . θin/x1 . . . xn] (1 ≤ i ≤ m)

Γ ⊢ match M in t x1 . . . xn return τ with c1 x11 . . . x1k1 ⇒ M1 | . . . | cm xm1 . . . xmkm ⇒ Mm end

: τ [b1, . . . , bn/x1 . . . xn]

(* 偶数の定義 *)

Inductive even : nat -> Prop :=

| even_O : even O

| even_SS : forall n, even n -> even (S (S n)).

(* 帰納的述語を証明する定理 *)

Theorem even_double n : even (n + n).

Proof.

elim: n => /= [|n IH].

+ apply: even_O.

+ rewrite -plus_n_Sm.

by apply: even_SS.

Qed.

(* 帰納的述語に対する帰納法もできる *)

Theorem even_plus m n : even m -> even n -> even (m + n).

Proof.

elim: m => //=.

Restart.

move=> Hm Hn.

elim: Hm => //= m m IH.

apply: even_SS.

Qed.

(* 矛盾を導き出す *)

Theorem one_not_even : ~ even 1.

Proof.

case.

Restart.

move H: 1 => one He. (* move H: exp => pat は H: exp = patを作る *)

case: He H => //.

Restart.

move=> He.

inversion He.

Show Proof. (* 証明が複雑で、SSReflectでは様々な理由で避ける *)

Qed.

(* 等式を導き出す *)

Theorem eq_pred m n : S m = S n -> m = n.

Proof.

move=> []. (* 等式を分解する *)

done.

Qed.

実は Coqの論理結合子のほとんどが帰納的述語として定義されている．

Inductive and (A B : Prop) : Prop := conj : A -> B -> A /\ B.

Inductive or (A B : Prop) : Prop :=

or_introl : A -> A \/ B | or_intror : B -> A \/ B.

Inductive ex (A : Type) (P : A -> Prop) : Prop :=

ex_intro : forall x : A, P x -> exists x, P x.

Inductive False : Prop := .

and，orや exについて caseが使えた理由がこの定義方法である．
しかも，Falseは最初からあるものではなく，構成子のない述語として定義されている．生成

される帰納法の補題をみると面白い．
Print False_ind.

fun (P : Prop) (f : False) => match f return P with end

: forall P : Prop, False -> P

ちょうど，矛盾の規則に対応している．作戦　 elimでそれが使える．
Theorem contradict (P Q : Prop) : P -> ~P -> Q.

Proof.

move=> p.

elim.

exact: p.

Qed.

練習問題 1.1 以下の定理を証明しなさい．
Module Odd.

Inductive odd : nat -> Prop :=

| odd_1 : odd 1

| odd_SS : forall n, odd n -> odd (S (S n)).

Theorem even_odd n : even n -> odd (S n). Abort.

Theorem odd_even n : odd n -> even (S n). Abort.

Theorem even_not_odd n : even n -> ~odd n. Abort.

End Odd.

2 MathCompのライブラリ
先週は ssreflectのコマンドを見たが，MathCompの本当の強さはそのライブラリにある．その

大きな特徴は書き換えを証明の基本手法とすること．
ライブラリは ssreflect, fingroup, algebra等，いくつかのの部分からできている．前者は

一般的なデータ構造で，後者は代数系の証明に使う．

Search

Ssreflect の Search コマンドが強力で，ライブラリを探すのに便利．
Search "add". (* 名前に add を含む定理を検索する *)

Search (_ + S _). (* 結論がパターンを含む定理を検索する *)

Search _ (_ + S _). (* 前提または結論がパターンを含む定理を検索する *)

Search (_ + _) (_ * _) "mul". (* 左を全てみたすものを検索する *)

基本データ
まず，ssreflectを読み込む．それほど多くはない．
From mathcomp Require Import all_ssreflect.

いくつかのモジュールが読み込まれます．ssrbool は論理式と述語の扱い．ssrnat は自然数．
ssrfun は関数 (写像)の様々な性質．seqはリスト．eqtype, choice, fintypeはそれぞれ等価
性，選択，有限性が使えるデータ構造のための枠組みを提供している．例えば，自然数の等価性
は判定できるので，排中律を仮定しなくても場合分けができる．
中身について，ファイルを参照するしかないが，まず ssrnatの例をみよう．(ちなみに，ソー

スファイルはは~/.local/share/coq/mathcomp/ssreflectの下にある)

4



Module Test_ssrnat.

Fixpoint sum n :=

if n is m.+1 then n + sum m else 0.

Theorem double_sum n : 2 * sum n = n * n.+1.

Proof.

elim: n => [|n IHn] //=.

rewrite -[n.+2]addn2 !mulnDr.

rewrite addnC !(mulnC n.+1).

by rewrite IHn.

Qed.

End Test_ssrnat.

自己反映
論理式も書き換えで処理したい．そのために，ssrboolでは論理演算子を型 boolの上の演算

子として定義している．例えば，/\は&&, \/は||になる．二つの定義の間に行き来するために，
reflectという自己反映を表した宣言を使う．それが SSReflectの名前の由来である．
Print reflect.

Inductive reflect (P : Prop) : bool -> Set :=

ReflectT : P -> reflect P true | ReflectF : ~ P -> reflect P false

Check orP.

orP : forall b1 b2 : bool, reflect (b1 b2) (b1 || b2)

表現の切り替えはビュー機構によって行われる．前に見た適用パターンを使う．move, case,

applyなどの直後に/viewを付けると，対処が可能な方向に変換される．=>の右でも使える．な
お，ビューとしては上の reflect型 だででなく，同値関係 (P <-> Q)や普通の定理 (P -> Q)も
使える．
Module Test_ssrbool.

Variables a b c : bool.

Print andb.

Lemma andb_intro : a -> b -> a && b.

Proof.

move=> a b.

rewrite a.

move=> /=.

done.

Restart.

by move ->.

Qed.

Lemma andbC : a && b -> b && a.

Proof.

case: a => /=.

by rewrite andbT.

done.

Restart.

by case: a => //= ->.

Restart.

by case: a; case: b.

Qed.

Lemma orbC : a || b -> b || a.

Proof.

case: a => /=.

by rewrite orbT.

by rewrite orbF.

Restart.

move/orP => H.

apply/orP.

move: H => [Ha|Hb].

by right.

by left.

Restart.

by case: a; case: b.

Qed.

Lemma test_if x : if x == 3 then x*x == 9 else x !=3.

Proof.

case Hx: (x == 3).

by rewrite (eqP Hx).

done.

Restart.

case: ifP.

by move/eqP ->.

move/negbT. done.

Qed.

End Test_ssrbool.

自己反映があると自然数の証明もスムーズになる．

Theorem avg_prod2 m n p : m+n = p+p -> (p - n) * (p - m) = 0.

Proof.

move=> Hmn.

have Hp0 q: p <= q -> p-q = 0.

by rewrite -subn_eq0 => /eqP.

suff /orP[Hpm|Hpn]: (p <= m) || (p <= n).

+ by rewrite (Hp0 m).

+ by rewrite (Hp0 n).

case: (leqP p m) => Hpm //=.

case: (leqP p n) => Hpn //=.

suff: m + n < p + p.

by rewrite Hmn ltnn.

by rewrite -addnS leq_add // ltnW.

Qed.

練習問題 2.1 以下の等式を証明しなさい．タクティクは rewriteのみでできる．
ssrnat_doc.v の補題でほぼ足りるが, leq_mulも便利．
Module Equalities.

Theorem square_sum a b : (a + b)^2 = a^2 + 2 * a * b + b^2. Abort.

Theorem diff_square m n : m >= n -> m^2 - n^2 = (m+n) * (m-n). Abort.

Theorem square_diff m n : m >= n -> (m-n)^2 = m^2 + n^2 - 2 * m * n. Abort.

End Equalities.

